CSE 378

Quiz Section #2

Coding in MIPS Assembly

Suppose that we want to write a simple program that takes in a number, sets each of the 100 elements of some global array to that number, and then prints out how many elements were actually set, to make sure. How would we go about doing this? Well, first, let’s write a quick C program that does this for us, and then slowly transform it into assembly.

int arr[100];

void main () {


int i, j;


scanf(“%d”, &j);


for (i = 0; i < 100; i++) {



arr[i] = j;


}


printf(“done: %d”, i);

}

So, now that we have our program, let us convert it to something that would be a bit easier to translate into assembly. Here is the reworked version:

int arr[100];

void main () {


int i, j;


scanf(“%d”, &j);


i = 0;


while (i < 100) {



arr[i] = j;



i = i + 1;


}


printf(“done: “);


printf(“%d”, i);

}

First of all, let us specify (to SPIM) that all this code is within the main function. Note the use of the .text directive. Everything that comes after it will be the actual executable code, and will get placed into memory as the “text” segment of the program. Note that ALL the code goes into this section. So, our function will now look like this:

int arr[100];

.text

.globl main

main:

int i, j;

scanf(“%d”, &j);

i = 0;

while (i < 100) {


arr[i] = j;


i = i + 1;

}

printf(“done: “);

printf(“%d”, i);

.end main

Alright, so now that we have clearly stated what our code is, let’s worry about that global array. To declare the array, we need to use the .data directive. This directive specifies to the assembler that space for some data needs to be allocated in the “static data” portion of the program memory. Well, great, but how do we actually declare what data it is? Three keywords are helpful:

.asciiz “hello” # reserves space for this string, and places



    # ‘hello’ into it

.space 50
    # reserves space for 50 bytes, for general use

.word 17
    # reserves space for a word (4 bytes), and



    # initializes it to 17

Note that ALL global variables, and NO non-global variables go into the “static data” section. A local variable, such as i or j, does not belong here. So, for our global array, we can simply use the “space” keyword to reserve enough memory. Note that our array is of size 100 ints, so we need to reserve 100 x 4 = 400 bytes of space. Also, note that the string “done: ” in the first printf() can also be placed into the global data section, using the “.asciiz” keyword. Making the appropriate changes, our program now looks like this…

.data

arr:
.space 400

msg:
.asciiz “done: “

.text

.globl main

main:

int i, j;

scanf(“%d”, &j);

i = 0;

while (i < 100) {


arr[i] = j;


i = i + 1;

}

printf(msg);

printf(“%d”, i);

.end main

Well, what about the local variables, like i and j? Where would they go? Simple – we can keep them in temporary-value registers ($t0 through $t7, or $8 through $15). One important rule to always adhere to: when programming in assembly, always always write down what each register represents. If you don’t, careless errors WILL happen, and painful debugging will ensue. So, let us start assigning some registers to specific tasks. $t0 will store i. $t1 will hold an address into the array. $t2 will hold the constant 100 – we need to check when to break out of the while loop, after all. $t3 will hold j. $t4 will be used for other purposes. Great, we’re set! Let’s initialize these variables to their necessary values:

.data

arr:
.space 400

msg:
.asciiz “done: “

.text

.globl main

main:

la $t1, arr

# Note how we initialized t1 to point to the




# beginning of our array

ori $t2, $0, 100
# Initializing t2 to its constant value

add $t0, $0, $0
# Initializing our incremental value

scanf(“%d”, &j);

while (i < 100) {


arr[i] = j;


i = i + 1;

}

printf(msg);

printf(“%d”, i);

.end main

Notice how we initialized $t1 by using the “arr” label – the (base) address of our array. Also of important note: when I learned this, we used addi $t1, $gp, arr to initialize $t1. This worked because $gp pointed to the beginning of the “static data” section of the program, and arr gave an offset from that. SPIM, however, dislikes this. So simply keep this in mind, should you ever need to take this into consideration. Now, let us write the loop in assembly. Remember, that we will want to break out of it when $t0 is no longer less than $t2. Also, at the very end of the function, we will want to return back to the caller. Therefore, we will need to jump to where we were called from – information stored in $ra.

.data

arr:
.space 400

msg:
.asciiz “done: “

.text

.globl main

main:

la $t1, arr

# Initializations

ori $t2, $0, 100

add $t0, $0, $0

scanf(“%d”, &j);

loopstart:

slt $t4, $t0, $t2

beq $t4, $0, done
# Breaking out of the loop

sw $t3, 0($t1)
# Writing one value into the array

addi $t0, $t0, 1
# Incrementing our loop counter (i) by 1

addi $t1, $t1, 4
# Going to the next spot in the array

j loopstart

# Going back up for another iteration

done:

printf(msg);

printf(“%d”, i);

jr $ra

# Returning back to the caller

.end main

Alright, and what about those pesky printfs and scanfs? How do we get rid of those? We use the syscall instruction to perform these operations. To do so, we insert the system call number into register $v0, and its arguments into the $a- registers. And then we issue the “syscall” instruction. More specifically, to print an integer, we use system-call number 1, and set $a0 to the integer we want to print. To print a string, we use system-call number 4, and set $a0 to contain the base address of the string. To read in an integer, we use system-call number 5. The result is returned in register $v0. These are far from the only system calls available – to find out more about them, consult the full listings in the book. So, after applying these changes to our program, it becomes…

.data

arr:
.space 400

msg:
.asciiz “done: “

.text

.globl main

main:

la $t1, arr

# Initializations

ori $t2, $0, 100

add $t0, $0, $0

addi $v0, $0, 5
# Reading in the value for $t3 (j)

syscall

add $t3, $v0, $0

loopstart:

slt $t4, $t0, $t2

beq $t4, $0, done
# Breaking out of the loop

sw $t3, 0($t1)
# Writing one value into the array

addi $t0, $t0, 1
# Incrementing our loop counter (i) by 1

addi $t1, $t1, 4
# Going to the next spot in the array

j loopstart

# Going back up for another iteration

done:

addi $v0, $0, 4
# Printing the result string

la $a0, msg

syscall

addi $v0, $0, 1
# Printing the result itself ($t0)

add $a0, $t0, $0

syscall

jr $ra

# Returning back to the caller

.end main

And this is the full assembly version of our simple C program above.

Now, in groups, try to write an assembly program that continually asks the user for a number to input. If that number is greater than zero, then the program computes the factorial of that number, prints the result to the screen, and asks the user again. Use the simple while (N != 0) {result = result * N; N--; } way of computing the factorial. If the number entered is zero or less, then the program exits. This is a rather tough problem, but with a systematic approach you can solve it. Once you are done, I suggest trying to run it within the SPIM simulator, to see if it works. :) Though this is not a required exercise, it will give you a great start on the homework.

